In-class Demonstration: Elementary Matrices

By Russell Blyth.

```
> restart: with(LinearAlgebra):
> randomize(3967);
```

$$3967 \quad (1)$$

```
>
```

Elementary row and column operations

Generate a random 3 x 4 matrix.

```
> A := RandomMatrix(3,4,generator=rand(-5..5));
```

$$A := \begin{bmatrix} -62 & 5 & 35 & -69 \\ -32 & 73 & 44 & -31 \\ 93 & 8 & -10 & 94 \end{bmatrix} \quad (1.1)$$

Also generate the 3x3 and 4x4 identity matrices for later use.

```
> I3 := IdentityMatrix(3);
  I4 := IdentityMatrix(4);
```

$$I3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (1.2)$$

$$I4 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Apply several elementary row and column operations.
The three elementary row operations can be performed using the RowOperation command in the LinearAlgebra package of Maple.
The command
RowOperation(M, [r2, r1], scal);
is used to add scal times row r1 of matrix M to row r2.
The command
RowOperation(M, r1, scal);
is used to multiply row r1 of M by scal. The command
RowOperation(M, [r1, r2]);
is used to switch rows r1 and r2 of the matrix M.

ColumnOperation() performs similar operations on columns.

We apply several elementary row operations to the matrix A; at the same time we compute the corresponding elementary matrices by applying the same row operations to the identity matrix. We can then check that multiplying A on the left by the elementary matrix produces the same result as the row operation.

```
> RowOperation(A,[1,2]);
  E1 := RowOperation(I3,[1,2]);
  E1.A;
```

$$\begin{bmatrix} 5 & -2 & 0 & -1 \\ 1 & 4 & 0 & -4 \\ 5 & -5 & 2 & -1 \end{bmatrix} \quad (1.3)$$

$$E1 := \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 5 & -2 & 0 & -1 \\ 1 & 4 & 0 & -4 \\ 5 & -5 & 2 & -1 \end{bmatrix}$$

```
> RowOperation(A,3,1/3);
  E2 := RowOperation(I3,3,1/3);
  E2.A;
```

$$\begin{bmatrix} 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ \frac{5}{3} & \frac{-5}{3} & \frac{2}{3} & \frac{-1}{3} \end{bmatrix} \quad (1.4)$$

$$E2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & \frac{1}{3} \end{bmatrix}$$

$$\begin{bmatrix} 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ \frac{5}{3} & \frac{-5}{3} & \frac{2}{3} & \frac{-1}{3} \end{bmatrix}$$

```
> RowOperation(A,[3,1],-1);
  E3 := RowOperation(I3,[3,1],-1);
  E3.A;
```

$$\begin{bmatrix} 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ 4 & -9 & 2 & 3 \end{bmatrix} \quad (1.5)$$

$$E3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ 4 & -9 & 2 & 3 \end{bmatrix}$$

```
> ColumnOperation(A,[2,4]);
  E4 := ColumnOperation(I4,[2,4]);
  A.E4;
```

$$\begin{bmatrix} 1 & -4 & 0 & 4 \\ 5 & -1 & 0 & -2 \\ 5 & -1 & 2 & -5 \end{bmatrix} \tag{1.6}$$

$$E4 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 & -4 & 0 & 4 \\ 5 & -1 & 0 & -2 \\ 5 & -1 & 2 & -5 \end{bmatrix}$$

```
> ColumnOperation(A,1,4);
  E5 := ColumnOperation(I4,1,4);
  A.E5;
```

$$\begin{bmatrix} 4 & 4 & 0 & -4 \\ 20 & -2 & 0 & -1 \\ 20 & -5 & 2 & -1 \end{bmatrix} \tag{1.7}$$

$$E5 := \begin{bmatrix} 4 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 4 & 4 & 0 & -4 \\ 20 & -2 & 0 & -1 \\ 20 & -5 & 2 & -1 \end{bmatrix}$$

```
> ColumnOperation(A,[4,1],-2);
  E6 := ColumnOperation(I4,[4,1],-2);
  A.E6;
```

$$\begin{bmatrix} 1 & 4 & 0 & -6 \\ 5 & -2 & 0 & -11 \\ 5 & -5 & 2 & -11 \end{bmatrix} \tag{1.8}$$

$$E6 := \begin{bmatrix} 1 & 0 & 0 & -2 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 4 & 0 & -6 \\ 5 & -2 & 0 & -11 \\ 5 & -5 & 2 & -11 \end{bmatrix}$$

Note that multiplication by an elementary matrix performs the corresponding elementary row or column operation. For a row operation the elementary matrix is multiplied on the left, while for a column operation the elementary matrix is multiplied on the right.

Now check the inverses of each of the elementary matrices we generated.

```
> E1; E1^(-1);
```

$$\begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad (1.9)$$

$$\begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> E2; E2^(-1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & \frac{1}{3} \end{bmatrix} \qquad (1.10)$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

```
> E3; E3^(-1);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -1 & 0 & 1 \end{bmatrix} \qquad (1.11)$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 1 & 0 & 1 \end{bmatrix}$$

```
> E4; E4^(-1);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix} \quad (1.12)$$

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

```
> E5; E5^(-1);
```

$$\begin{bmatrix} 4 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.13)$$

$$\begin{bmatrix} \frac{1}{4} & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> E6; E6^(-1);
```

$$\begin{bmatrix} 1 & 0 & 0 & -2 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad (1.14)$$

$$\begin{bmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Note that each elementary matrix is invertible, and that it's inverse is the elementary matrix corresponding to the elementary row or column operation that undoes the elementary row or column operation corresponding to the original elementary matrix.

```
>
```